
1

Paper 1342-2014
Extreme SAS® reporting II: Data Compendium and 5 Star Ratings Revisited

Louise S. Hadden, Abt Associates Inc., Cambridge, MA

ABSTRACT

Each month, our project team delivers updated 5-Star ratings for 15,700+ nursing homes across the United States to
Centers for Medicare & Medicaid Services. There is a wealth of data (and processing) behind the ratings, and this
data is longitudinal in nature. A prior paper in this series, "Programming the Provider Previews: Extreme SAS
Reporting" discussed one aspect of the processing involved in maintaining the Nursing Home Compare website. This
paper will discuss two other aspects of our processing: creating an annual data Compendium, and extending the 5-
star processing to accommodate several different output formats for different purposes. Products used include Base
SAS, SAS/STAT, SG Procedures, and SAS/Graph. New annotate facilities in both SAS/Graph and the SG
Procedures will be discussed. This paper and presentation will be of most interest to SAS programmers with medium
to advanced SAS skills.

INTRODUCTION

Since 1998, the U.S. Centers for Medicare and Medicaid Services (CMS) has maintained a website, Nursing Home
Compare, which provides detailed quality information about every certified nursing home in the country. In December
2008, CMS greatly enhanced the usability of the website by adding an easy-to-understand 5-star rating. Each nursing
home receives one to five stars based on performance in each of three key quality domains (health inspections,
reported staffing levels, and quality measures derived from mandated assessments of resident health and well-being)
plus an overall quality rating. Calculation of ratings requires integration of information from both facility and resident-
level data sources. SAS® was used extensively in analysis to support the development of the rating system, and it is
currently used to process data to refresh the ratings each month, based on newly collected data in each domain. It
has evolved over time, but the general idea remains the same. The rating process is described in a prior paper,
“Measuring Nursing Home Quality – The Five-Star Rating System”, presented at SAS Global Forum in 2010.

A preview of the month’s ratings in the form of a customized three-page PDF report is generated for each provider for
each month. These reports are automatically loaded into providers’ e-mailboxes by means of specific identifying
information embedded in the file name for each provider’s report. The provider previews were the subject of an
earlier paper, and have evolved steadily throughout the five years we have been producing them. The data sources
for the reports now include the Provider Rating file created at the conclusion of data processing for the month, Quality
Measure data, inspection data, staffing data and ownership data. If a provider’s ratings put them at risk for being a
“special focus facility”, a warning paragraph is conditionally printed on the first page of the preview. Information on
any changes to the 5 star rating system, Nursing Home Compare and/or the previews is also printed on the first page.
Although the previews have evolved over time, the process is described in a prior paper, "Programming the Provider
Previews: Extreme SAS Reporting", presented at NESUG 2011 and SAS Global Forum 2012.

Data refreshes for the Nursing Home Compare web site were originally done in the form of text files generated by
data _null_ and put statements. These files were zipped and transferred to project staff at the Centers for Medicare &
Medicaid Services, processed further, then transferred to another vendor who again processed the data further
before posting on the Nursing Home Compare website each month. In order to calculate ratings and report on our
processing, including the provider preview reports, we were already doing much of what CMS and the vendors were
doing. To streamline the process, improve consistency and reduce redundant efforts, we undertook an effort to begin
generating “web-ready” output for delivery to CMS and its vendors.

In addition to the original output files and provider previews, we now provide:

 Analytic Reports (including state specific ratings spreadsheets containing the ratings for each provider in the
state, a ratings spreadsheet for all providers in the U.S., state specific quality measure (QM) spreadsheets
containing quality measures for each provider in the state, a QM spreadsheet for all providers in the U.S.,
staffing details for all providers in the U.S., and assorted reports on thresholds and the change in ratings
from month to month.)

 Helpline Database and Materials (including a Microsoft Access data base to assist the call center with
questions providers might have, materials to produce a Section 508 compliant version of the Provider
Previews, and sample provider previews.)

ods escapechar='^';

proc print data = composite label
style(header) = {font=("Arial",10pt,Bold) just=center vjust=bottom
background=lightblue}
style(obsheader) = {font=("Arial",10pt,Bold) just=left vjust=bottom
background=lightblue}
style(data) = {font=("Arial",10pt,Normal) background=grayee} ;
where state = "&state" ;

id provnum / style(header data)={vjust=bottom font_style=italic
htmlstyle= "VND.MS-EXCEL.NUMBERFORMAT:@"
cellwidth=.75in };

var provname
/ style(header data)={vjust=bottom just=left cellwidth=3in } ;

var overall health mdsqm staffing rnstaff
/ style(header data)={vjust=bottom just=center cellwidth=1.25in };

var defscore
/ style(header data)={vjust=bottom just=center

htmlstyle= "vnd.ms-excel.numberformat:0.00"}; . . .

The code snippet and screen shot below are for an ODS TAGSETS.EXCELXP (XML output) table created using
PROC TABULATE. Notice the Unicode stars!

proc tabulate data=composite . . . ;
class &clsvar statename

/ s={bordercolor=black borderwidth=1
font=("Helvetica",9pt,Normal)};

classlev statename /
s={bordercolor=black borderwidth=1
font=("Helvetica",9pt,Normal)};

classlev &clsvar. / s = {font=("Helvetica",9pt,Normal) background=white } ;
keyword all / s = {font=("Helvetica",9pt,bold)};

tables(all={label='All States'
s={font=("Helvetica",9pt,Bold)}}*{s={font=("Helvetica",9pt,Bold)}}

statename=' '),
(all={label='TOTAL N^{super 1}'
s={font=("Helvetica",9pt,Bold)}}*{s={font=("Helvetica",9pt,Bold)}}*
n=' '*f=comma6.*{s={font=("Helvetica",9pt,Normal)
tagattr='format:#,###'}}
&clsvar*(n*f=comma5.*{s={font=("Helvetica",9pt,Normal) just=right
tagattr='format:#,###'}}
pctn<&clsvar>*f=6.2*{s={font=("Helvetica",9pt,Normal) just=right
tagattr='format:##0.00'}} ) )
/ printmiss misstext='0';

keylabel pctn ='%' ;
keyword all / s={font=("Helvetica",9pt,bold) just=left};
keyword pctn / s={font=("Helvetica",9pt,bold) just=left};
keyword N / s={font=("Helvetica",9pt,bold) just=right};
format &clsvar starfmt. ;

You may have noticed a slight problem. These outputs are native HTML and XML, respectively, not Excel. If we
open the output in Excel (Office 2007 and later) we get the dreaded warning.

Since these outputs are delivered to a client, we did not want the warnings to appear, and we also did not want to
manually open and “save as” the outputs as there are over 100 workbooks generated each month. We use DDE to
“save as” and freeze column and row headers in a process described in a Visual Displays paper and presentation,
“EXCELing in DDE: Unlock Useful Tools for Processing Excel®”. The outputs are used by a variety of end-users.

Helpline Materials

The provider previews are delivered to providers via electronic mailboxes the week before the new ratings are posted
each month on the third Thursday. The helpline is open for the week including the third Thursday, so that providers

can alert the helpline and CMS to any problems with the data presented in the previews. The helpline materials are
relevant to a paper on reporting in that they highlight the need to have a Section 508 compliance solution for all
materials produced for the government and other entities. The provider previews are produced by SAS in PDF
format, which is not “tagged” PDF which is required for Section 508 compliance at this point. Since there are over
15,500 provider previews generated each month, it isn’t possible to open and “tag” each preview in third-party
software. An access data base is provided with all the data provided on the provider previews along with a template
to merge the data into that mimics the appearance of the provider preview for any providers who need to view the
provider preview with a reader. The access data base also allows Helpline staff to do look-ups if providers have
questions.

Statements of Deficiencies (2567s)

In order to quality for participation in the Medicare and Medicaid programs, nursing homes have to meet certain
standards. State governments do health and fire safety inspections of eligible nursing homes and investigate any
complaints. CMS has entered into an agreement with state governments to do health inspections and fire safety
inspections of these nursing homes and investigate complaints about nursing home care. The standards cover a wide
range of topics, from proper management of medications, protecting residents from physical or mental abuse and
inadequate care, to the safe storage and preparation of food. Inspections take place about once a year, or more
frequently in the case of poor performance. Deficiency citations are issued if a nursing home does not meet a
standard. These citations are recorded on a detailed inspection report, form HCFA-2567. One of the elements of the
2567 is a 32,767 byte long free text field including PII and embedded non-printing characters.

CMS publishes 2567s in their entirety on Nursing Home Compare. In order to protect privacy of nursing home
residents and staff, the free text field must be redacted. Part of our processing routine for each month includes
running these very large files and the very large text field through a redaction process using Perl regular expressions
to replace sensitive information with generic statements in brackets.

Reporting on the 2567s presents additional challenges due to the very long text field. We provide the 2567s in three
formats: Tab-delimited (discussed in the next section), Excel, and Microsoft Access (converted to SQL server data
base.) The redacted 2567 SAS data set (with both the redacted and un-redacted text field) is approximately 10
gigabytes and has approximately 300,000 records. As such, it is too large to export directly. Removing the
unredacted text field reduces the file size considerably, but it is still too large to export in its entirety. For the Excel
output, we split the file into 10 CMS regions and report on a subset of variables using PROC EXPORT. For the
Access output, we split the file into 2 files per month of inspection date, resulting in 24 files, again with a (different)
subset of variables, and reported on using PROC EXPORT. Access tables are then loaded into a SQL server data
base (for a variety of reasons we cannot directly export the large SAS data base to SQL server.)



6

Tab-delimited Output

Prior to July 2013, flat text files were being delivered to CMS, who pre-processed the data files and then passed them
onto a vendor, who again processed the data in order to refresh the web site. CMS, the vendor and Abt Associates
collaborated to deliver the vendor data that could be uploaded to the web site with less duplicative effort. Although
the intent is to move to XML in the future, tab-delimited output arranged in a XML-like structure is the current delivery
mechanism. In most cases, we provided the pre-processing to arrange the data correctly, and SAS delivered with
PROC EXPORT. However, our friends the 2567s begged to differ. There is a 32767 record length limit on delimited
output files created using PROC EXPORT. We wrote an “old style” data _null_ and put routine using hard-coded hex
tabs in between variables with a longer record length and thought it would be perfect. Unfortunately, the embedded
tabs, carriage returns and other “stuff” in the redacted text field wreaked havoc. Even worse, we needed to retain line
breaks because multiple deficiencies could be addressed in a single 2567 report and line breaks were the
differentiation between deficiencies. We stripped embedded tabs and used the TRANWRD function to convert
carriage returns and line breaks to the HTML code for line breaks, and that worked!

Data.Medicare.Gov

There are a number of other compare tools similar to Nursing Home Compare on the Medicare.Gov website:
Hospital Compare, Home Health Compare and Dialysis Facility Compare. All of the data that power these web pages
is available for download and online access at Data.Medicare.Gov. As of July, we are also providing the Nursing
Home Compare data files for Data.Medicare.Gov for minimal pre-processing and upload. This task presented a
different challenge. Data files were to be constructed as CSV files, except that more descriptive variable / column
names were desired and we needed to be able to preserve formatting in variables (provider number, zip code, etc.)
PROC EXPORT to CSV would not work for either of these issues. ODS TAGSETS.CSV did allow us to preserve our
formatting by providing the quote by type option. Re-reading the data set in and outputting, replacing variable names
with comma delimited text names in row 1 gave us more descriptive column headers. Note that for a large number of
variables you should use an appropriate LRECL statement on both the file and infile statements, or your rows will be
truncated and there will not be any warning message! (I learned this the hard way.)

ods csv file=".\outfiles\intermediate\penalties0.csv" options(quote_by_type="yes");

proc print data=penalties_csv noobs;
run;

ods csv close;

filename in5 ".\outfiles\intermediate\penalties0.csv";
filename out5 ".\outfiles\penalties.csv";

data test;
line='"Federal Provider Number",

"Provider Name",
"Penalty Date",
"Penalty Type",
"Fine Amount",
"Payment Denial Start Date",
"Payment Denial Length in Days",
"Provider Address",
"Provider City",
"Provider State",
"Provider Zip Code",
"Location",
"Processing Date"';

file out5;
if _n_ = 1 then put line;
infile in5 dlm='0D'x firstobs=2;
input ;
put _infile_;

run;

Our final choices reflected our desire to not overwhelm the viewer with too much data, as can be seen in the before
and afters presented below.

It was not possible to distinguish Deficiency levels H-L,
and we were unable to keep with our blue color scheme.

Additional years of data were added, and levels G-L were
collapsed to create a more informative figure.

Viewers felt that the narrow bars inside the wider bars
were confusing. In addition, we were unable to annotate
numbers on the narrow bars easily. The placement of
the legend was also not optimal.

We recreated the figure as a stacked bar chart, added
more years of data, and were able to annotate both
numbers in each bar as well as the percent occupied at
the top of each bar.

Although our mantra was “less is more”, there are also times when useful information can be conveyed by breaking
data into pieces for presentation. We use both of these graphs. The second graph shows the same information as
the first graph (number of nursing home facilities in the United States by year), but adds subcategories (number of
nursing home facilities by bed size by year).

The Data Compendium and Section 508 Compliance

A final production step incorporates all the separate RTF table and figure outputs into a single document, creates a
table of contents, adds a cover sheet, and prepares for converting to “tagged” PDF suitable for Section 508
compliance. While the tabular outputs are easily converted to tagged PDF, figures need “alt text” which describe the
figure for reading software. The same data that goes into the figures can be used to generate “alt text” for each figure
programmatically. We take advantage of the same techniques used to create the sentences in the provider preview
to do this. We use macro processing as this is used for multiple figures and will be reused for additional years.

%macro gen508(fignum,figvar,figtit,figyear,lctit,outfi,source,infi);

proc freq data=dd.&infi._&figyear.;
tables &figvar._5*legend_label / missing list out=tryit;

run; /* sorts omitted from code snippet */

proc transpose data=dd.&infi._&figyear. (keep=state_name legend_label)
out=tryit2 prefix=st_;
var state_name;
by legend_label;

run;

data tryit3;
length range $ 32 numword $ 6 y $ 2000 ;
merge tryit tryit2;
by legend_label;

x=compress(reverse(substr(left(reverse(legend_label)),1,4)),'()');
lenrange=length(legend_label);
if x='9' then y=catx(', ',st_1,st_2,st_3,st_4,st_5,st_6,st_7,st_8,'and',st_9);
if x='10' then y=catx(',

',st_1,st_2,st_3,st_4,st_5,st_6,st_7,st_8,st_9,'and',st_10);
if x='11' then y=catx(',

',st_1,st_2,st_3,st_4,st_5,st_6,st_7,st_8,st_9,st_10,'and',st_11);
y=tranwrd(y,' and,',' and');
y=catt(y,'.');
legend_label=compbl(legend_label);
revleg=left(reverse(legend_label));
subrev=substr(revleg,1,4);
if subrev=')9( ' then range=reverse(substr(revleg,5,lenrange-4));
if subrev in(')01(',')11(') then range=reverse(substr(revleg,6,lenrange-5));
range=left(range);
drop revleg subrev st_: _: ;
if x='9' then numword='Nine';
if x='10' then numword='Ten';
if x='11' then numword='Eleven';

run;

data sent1 (rename=(y=sent1 numword=num1 range=range1 ))
sent2 (rename=(y=sent2 numword=num2 range=range2))
sent3 (rename=(y=sent3 numword=num3 range=range3))
sent4 (rename=(y=sent4 numword=num4 range=range4))
sent5 (rename=(y=sent5 numword=num5 range=range5));

set tryit3 (keep=&figvar._5 y numword range);
if &figvar._5=0 then output sent1;
if &figvar._5=1 then output sent2;
if &figvar._5=2 then output sent3;
if &figvar._5=3 then output sent4;
if &figvar._5=4 then output sent5;

run;

data tryit4;
length blurb blurb1 blurb2 blurb3 blurb4 blurb5 $ 20000 figtit lctit source $ 500;
merge sent1-sent5;
lctit="&lctit";

fignum="&fignum";
figtit="&figtit";
source=cats('^n^n',"&source.");
constant_text1='Figure';
constant_text2=' is a five color map of the United States that displays the';
constant_text3='states have';
constant_text4='for each state.';
constant_text5='Guam, Puerto Rico and the United States Virgin Islands are not

included in this figure.';
blurb1=catx(' ',num1,constant_text3,range1,lctit,sent1);
blurb2=catx(' ',num2,constant_text3,range2,lctit,sent2);
blurb3=catx(' ',num3,constant_text3,range3,lctit,sent3);
blurb4=catx(' ',num4,constant_text3,range4,lctit,sent4);
blurb5=catx(' ',num5,constant_text3,range5,lctit,sent5);
blurb=catx('

',constant_text1,fignum,constant_text2,figtit,constant_text4,blurb1,blurb2,blurb3,
blurb4,blurb5,constant_text5,source);

blurb=compbl(blurb);
label blurb =" ";

run;

ods listing close;
ods escapechar='^';

ods rtf file="&outfi..rtf" path=odsout style=styles.noborder;
title2 "Figure &fignum. Section 508 compliance text";
proc report nowd data=tryit4

style(report)=[cellpadding=3pt vjust=b]
style(header)=[just=center font_face=Helvetica font_weight=bold font_size=10pt]
style(lines)=[just=left font_face=Helvetica] ;

columns blurb ;
define blurb / style(COLUMN)={just=l font_face=Helvetica

font_size=10pt cellwidth=1000 }style(HEADER)={just=l font_face=Helvetica
font_size=10pt };

run;
ods rtf close;
ods listing;
%mend;

%gen508(1.6,beds65p,Number of Certified Nursing Home Beds Per Thousand Persons Aged 65
Years and Older,2011,certified nursing home beds per thousand persons aged 65 years
and older including,fig1_6_508,The source of data for this figure is CASPER and the
U.S. Census.,fig1_6);

The Future of the Data Compendium

Two options we are looking into (with varying degrees of success) are embedding images (in this case “sparklines”
created with PROC SGPLOT) in procedural output. This works well with PROC REPORT in a table. Cynthia Zender
of SAS has now incorporated this process in her class on Reporting.

Although SAS has added the capability of annotating images in SAS/GRAPH output including maps, the only two
image styles are “tile” and “fit”. As you can see in the sample below, the results are less than attractive and
informative. Ideally we could have the ability to annotate small images at geographic centroids.

I’ve submitted an “official” request through Tech Support, and the issue is now on the SASWARE ballot. You can
vote via this link. Don’t be deterred by the not so informative description of the issue!

https://communities.sas.com/ideas/1178

If you are interested in trying out the PROC REPORT with sparklines technique, a zip file with code and sample data
is available at http://www.sascommunity.org/wiki/Behind_the_Scenes:_Using_Custom_Graphics_in_SAS.

CONCLUSION

Two very important aspects of our 5-Star Rating project are delivering and describing the data we process and create
each month, which is the very essence of reporting. Reporting on the wealth of data we process with several
different destinations and use cases is made easy by SAS. Among the options used are custom style templates,
annotate, macro processing, SAS/ACCESS for PC files, Unicode characters and other style functions, style
overrides, and ODS tagsets. Your reporting can be greatly enhanced by the use of these options.

ACKNOWLEDGMENTS

Robert Allison, Scott Huntley and Cynthia Zender of SAS provide me with help and inspiration in my varied SAS
explorations.

The author gratefully acknowledges the contributions to the Data Delivery and Compendium tasks in the Nursing
Home Compare 5-star project from:
Abt Associates Inc.: Christianna Williams, Allison Muma, Alan White, Alrick Edwards, Xin (Calvin) Guan, Patricia
Rowan and Michael Plotzke
Centers for Medicare and Medicaid Services: Edward Mortimore, Daniel Anderson, Jon Freedlander, Ian Kramer,
and Jean Scott.

REFERENCES AND RECOMMENDED READING

Robert Allison’s Examples! http://robslink.com/SAS/Home.htm

Allison, Robert. SAS/GRAPH: Beyond the Basics. SAS Institute, February 2012.



15

Guan, Calvin and Hadden, Louise S. “EXCELing in DDE: Unlock Useful Tools for Processing Excel®”.
Proceedings of NESUG 2013 Conference, September 2013.

Hadden, Louise S. “Behind the Scenes with SAS®: Using Custom Graphics in SAS® Output.” Proceedings of SAS
Global Forum 2013 Conference. April 2013.

Heath, Dan. “Now You Can Annotate Your Statistical Graphics Procedure Graphs.” Proceedings of SAS Global
Forum 2011 Conference, March 2011.

Huntley, Scott. “Let the ODS Printer Statement Take Your Output into the Twenty-First Century.” Proceedings of
the Thirty-First Annual SAS® Users Group International Conference. March 2006.

Huntley, Scott. “How to Add a Little Spice to your PDF Output.” Proceedings of SAS Global Forum 2008
Conference. March 2008.

Williams, Christianna and Hadden, Louise. “Programming the Provider Previews: Extreme SAS® Reporting”,
Proceedings of SAS Global Forum 2012 Conference, April 2012.

Williams, Christianna; Hadden, Louise; Mortimore, Edward; Nagy, Frank, Plotzke, Michael; and White, Alan.
“Measuring Nursing Home Quality – The Five-Star Rating System”. Proceedings of SAS Global Forum 2010
Conference. April 2010.

Zender, Cynthia L. and Kalt, Mike. “At the Crossroads: How to Decide on Your Graphics Path”. Proceedings of SAS
Global Forum 2012 Conference, April 2012.

Zender, Cynthia. “Creating Complex Reports.” Proceedings of SAS Global Forum 2008 Conference, March 2008.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Louise Hadden louise_hadden@abtassoc.com

Code samples are available upon request.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS
Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.


